Problem:
Score : 300 points

Problem Statement

You have a pot and N ingredients. Each ingredient has a real number parameter called value, and the value of the i-th ingredient (1 \leq i \leq N) is v_i.

When you put two ingredients in the pot, they will vanish and result in the formation of a new ingredient. The value of the new ingredient will be (x + y) / 2 where x and y are the values of the ingredients consumed, and you can put this ingredient again in the pot.

After you compose ingredients in this way N-1 times, you will end up with one ingredient. Find the maximum possible value of this ingredient.

Constraints

2 \leq N \leq 50

1 \leq v_i \leq 1000

All values in input are integers.

Input

Input is given from Standard Input in the following format:

N
v_1 v_2 \ldots v_N

Output

Print a decimal number (or an integer) representing the maximum possible value of the last ingredient remaining.

Your output will be judged correct when its absolute or relative error from the judge's output is at most 10^{-5}.

Sample Input 1

2
3 4

Sample Output 1

3.5

If you start with two ingredients, the only choice is to put both of them in the pot. The value of the ingredient resulting from the ingredients of values 3 and 4 is (3 + 4) / 2 = 3.5.

Printing 3.50001, 3.49999, and so on will also be accepted.

Sample Input 2

3
500 300 200

Sample Output 2

375

You start with three ingredients this time, and you can choose what to use in the first composition. There are three possible choices:

Use the ingredients of values 500 and 300 to produce an ingredient of value (500 + 300) / 2 = 400. The next composition will use this ingredient and the ingredient of value 200, resulting in an ingredient of value (400 + 200) / 2 = 300.

Use the ingredients of values 500 and 200 to produce an ingredient of value (500 + 200) / 2 = 350. The next composition will use this ingredient and the ingredient of value 300, resulting in an ingredient of value (350 + 300) / 2 = 325.

Use the ingredients of values 300 and 200 to produce an ingredient of value (300 + 200) / 2 = 250. The next composition will use this ingredient and the ingredient of value 500, resulting in an ingredient of value (250 + 500) / 2 = 375.

Thus, the maximum possible value of the last ingredient remaining is 375.

Printing 375.0 and so on will also be accepted.

Sample Input 3

5
138 138 138 138 138

Sample Output 3

138

Solution in Fortran (gfortran v4.8.4) (judge verdict: Accepted)

module ABC138

  ! <module>s to import
  use, intrinsic :: iso_fortran_env

  ! require all variables to be explicitly declared
  implicit none

  ! accessibility of <subroutine>s and <function>s in this <module>
  public  :: task_C
  private :: sort_heap

  ! contained <subroutine>s and <function>s are below
  contains

  subroutine task_C

    ! variables for this <subroutine>
    integer      :: num_ingredient
    real(REAL64) :: value_maximum

    ! support variables for this <subroutine>
    integer :: itr

    ! arrays for this <subroutine>
    real(REAL64), dimension(:), allocatable :: value_ingredient

    ! STEP.01
    ! read out the number of given ingredient
    read(unit=INPUT_UNIT, fmt=*) num_ingredient

    ! STEP.02
    ! allocate the array to store value of ingredients
    allocate( value_ingredient(1:num_ingredient) )

    ! STEP.03
    ! read out given value of ingredients
    read(unit=INPUT_UNIT, fmt=*) value_ingredient(:)

    ! STEP.04
    ! sort the array storing the value of ingredients
    call sort_heap ( value_ingredient(:), num_ingredient )

    ! STEP.05
    ! calculate the answer of this task
    value_maximum = value_ingredient(1)

    do itr = 2, num_ingredient, 1
      value_maximum = 0.5_REAL64 * (value_maximum + value_ingredient(itr))
    end do

    ! STEP.06
    ! output the answer of this task
    write(unit=OUTPUT_UNIT, fmt='(ES23.15e3)', advance='yes') value_maximum

    ! STEP.05
    ! deallocate the array to store the given data
    deallocate( value_ingredient )

    ! STEP.END
    return

  end subroutine task_C

  subroutine sort_heap (array, size)

    ! arguments for this <subroutine>
    real(REAL64), intent(inout) :: array(:)
    integer,      intent(in)    :: size

    ! variables for this <subroutine>
    integer      :: idx_p
    integer      :: idx_c
    integer      :: size_heap
    real(REAL64) :: buffer_val

    ! support variables for this <subroutine>
    integer :: itr

    size_heap = size     + 1
    itr       = size / 2 + 1
    
    do while (.true.)

      if (itr .gt. 1) then

        itr        = itr - 1
        buffer_val = array(itr)

      else

        size_heap = size_heap - 1
        if (size_heap .eq. 1) return
        buffer_val       = array(size_heap)
        array(size_heap) = array(1)

      end if
    
      idx_p = itr
      idx_c = itr * 2

      do while ( idx_c .lt. size_heap )

        if ( idx_c + 1 .lt. size_heap ) then
          if ( array(idx_c + 1) .gt. array(idx_c) ) idx_c = idx_c + 1
        end if

        if ( array(idx_c) .gt. buffer_val ) then
          array(idx_p) = array(idx_c)  
          idx_p        = idx_c         
          idx_c        = idx_p * 2     
        else
          exit 
        end if

      end do

      array(idx_p) = buffer_val

    end do

  end subroutine sort_heap

end module ABC138


program main

  ! <module>s to import
  use, non_intrinsic :: ABC138

  ! require all variables to be explicitly declared
  implicit none

  call task_C

end program main